
 PRÁCTICA 1. CURSO 2016-2017.
ANÁLISIS COMPLEJO.

 1 Operaciones básicas con números complejos

 Veremos incialmente algunos comandos específicos para trabajar con 
números complejos. Recordemos que en Maxima la unidad imaginaria se
representa por %i.

Podemos comenzar definiendo el número complejo z=3-2i:

(%i1) z:3-2*%i;

(%o1) 3-2 %i

 Si necesitamos escribir la parte real de z usaremos

(%i2) realpart(z);

(%o2) 3

 mientras que para la parte imaginaria

(%i3) imagpart(z);

(%o3) -2

 Sea entonces w=3+2i . Podemos operar con ambos complejos, y así hacer:

(%i4) w:-1+2*%i;

(%o4) 2 %i-1

 Notemos que a veces Maxima nos devuelve la parte imaginaria del complejo
antes que la parte real. Para que siempre escriba primero la parte
real seguida de la parte imaginaria, haremos:

(%i5) powerdisp:true$
w;

(%o6) -1+2 %i

(%i7) /*podemos calcular la suma o diferencia de ambos, 
que los introducimos como unas nuevas variables, 
que llamamos suma y resta*/

suma:z+w;
resta:z-w;

(%o7) 2

(%o8) 4-4 %i

 Podemos multiplicar complejos como se multiplican cualquier par de nros:

(%i9) z*w;

(%o9) ( )3-2 %i ( )-1+2 %i

 Observamos que nos da una expresión literal, por lo que si queremos 
calcular su resultado haremos:

(%i10) expand(%);

(%o10) 1+8 %i

 Lo mismo ocurre si intentamos dividir:

(%i11) z/w;

(%o11) 
3-2 %i

-1+2 %i

 y si lo que queremos es que nos de en forma cartesiana, haremos:

(%i12) rectform(z/w);

(%o12) -
7

5
-
4 %i

5

 Podemos calcular el módulo de z, usando abs(z) o cabs(z);:

(%i13) abs(z);

(%o13) 13

(%i14) cabs(z);

(%o14) 13

 NOTA:Si usamos abs(z) para calcular el módulo del complejo z, 
obtendremos el valor absoluto de z, y no el resultado dado por 
cabs(z), que nos da (x^2+y^2)^(1/2). La opción abs(z) suele 
usarse cuando se utilizan cálculos simbólicos (por ejemplo, 
calcular límites, derivadas, integrales,...)

 Para hallar el conjugado de z haremos:

(%i15) conjugate(z);

(%o15) 3+2 %i

 Podemos expresar el complejo z en forma polar |z|·e^(i·arg(z)):

(%i16) polarform(z);

(%o16) 13 %e
-%i atan





2

3

 Observamos que esta última forma nos da el módulo y el
argumento de z. No obstante, podemos calcular directamente
su argumento (anteriormente ya hemos calculado su módulo) sin
necesidad de expresarlo en forma polar. Para ello usaremos:

(%i17) carg(z);

(%o17) -atan







  
2

3

 Podemos calcular potencias naturales de z:

(%i18) z^2;
expand(%);

(%o18) ( )3-2 %i
2

(%o19) 5-12 %i

 También podemos calcular potencias enteras de z:

(%i20) z^(-2);
expand(%);
rectform(%);

(%o20) 
1

( )3-2 %i
2

(%o21) 
1

5-12 %i

(%o22) 
5

169
+
12 %i

169

 A la hora de calcular raíces complejas, hemos de tener en
cuenta que Maxima no las va a obtener todas (sabemos que todo
complejo z tiene n raíces n-simas), sino que nos va a dar
la primera de ellas (recordemos que las demás siempre tienen
el mismo módulo pero se diferencian en el argumento):

(%i23) z^(1/2);
rectform(%);

(%o23) 3-2 %i

(%o24) 
3+ 13

2
-

-3+ 13 %i

2

(%i25) z^(1/3);
rectform(%);

(%o25) ( )3-2 %i
1 3/

(%o26) 13
1 6/

cos









 
  

atan


  

2

3

3
-13

1 6/
%i sin









 
  

atan


  

2

3

3

 Podemos calcular la exponencial de z (es decir, e^z):

(%i27) exp(z);
rectform(%);

(%o27) %e
3-2 %i

(%o28) %e
3
cos( )2 -%e

3
%i sin( )2

 o el logaritmo principal de z:

(%i29) log(z);
rectform(%);

(%o29) log( )3-2 %i

(%o30) -%i atan







  
2

3
+
log( )13

2

 Podemos calcular senos y cosenos de z:

(%i31) cos(z);
rectform(%);

(%o31) cos( )-3+2 %i

(%o32) cosh( )2 cos( )3 +%i sinh( )2 sin( )3

(%i33) sin(z);
rectform(%);

(%o33) -sin( )-3+2 %i

(%o34) -%i sinh( )2 cos( )3 +cosh( )2 sin( )3

 Si preferimos la notación exponencial, "exponentialize()" escribe 
todo en términos de exponenciales:

(%i35) exponentialize(cos(z));

(%o35) 
%e

-%i( )-3+2 %i
+%e

%i( )-3+2 %i

2

 y "demoivre()" utiliza senos y cosenos en la salida en lugar de 
las exponenciales:

(%i36) demoivre(cos(z));

(%o36) cos( )-3+2 %i

 2 Ecuaciones

(%i37) /*Primero borramos todas las variables que hemos
usado anteriormente*/

remvalue(all);

(%o37) [z,w,suma,resta]

 En Maxima, una ecuación es una igualdad entre dos expresiones 
algebraicas escrita con el símbolo =.

(%i38) 3*x^2+2*x+x^3-x^2=4*x^2;

(%o38) 2 x+2 x
2
+x

3
=4 x

2

 Podemos asignarle un nombre para referirnos a esa ecuación:

(%i39) mi_ecuacion:3*x^2+2*x+x^3-x^2=4*x^2;

(%o39) 2 x+2 x
2
+x

3
=4 x

2

 lo que nos permite operar con ella:

(%i40) mi_ecuacion*4-3*x^3;

(%o40) -3 x
3
+4( )2 x+2 x

2
+x

3
=16 x

2
-3 x

3

  2.1 Resolución de ecuaciones y sistemas

 Maxima puede resolver los tipos más comunes de ecuaciones 
y sistemas de ecuaciones algebraicas de forma exacta. Por 
ejemplo, sabe encontrar las raíces de polinomios de grado 
bajo (2,3 y 4). En cuanto a polinomios de grado más alto 
o ecuaciones más complicadas, no siempre será posible
encontrar la solución exacta. En este caso, podemos 
intentar encontrar una solución aproximada en lugar de la 
solución exacta.

La orden "solve" nos da todas las soluciones, ya sean reales 
o complejas de una ecuación sistema de ecuaciones:

-->  remvalue(all);

(%i41) solve(x^2-3*x+1=0,x);

(%o41) [x=-
-3+ 5

2
,x=

3+ 5

2
]

 También podemos resolver ecuaciones que dependan de algún 
parámetro:

(%i42) solve(x^3-a*x^2-x^2+2*x=0,x);

(%o42) [x=-
-1-a+ -7+2 a+a

2

2
,x=

1+a+ -7+2 a+a
2

2
,x=0]

 Podemos obtener la multiplicidad de las raíces:

(%i43) solve(x^7-2*x^6+2*x^5-2*x^4+x^3=0);

(%o43) [x=-%i,x=%i,x=1,x=0]

(%i44) multiplicities;

(%o44) [1,1,2,3]

 La orden solve no sólo puede resolver ecuaciones algebraicas:

(%i45) solve(sin(x)*cos(x)=0,x);

solve: using arc-trig functions to get a solution.

Some solutions will be lost.

(%o45) [x=0,x=
π

2
]

 Como cualquiera puede imaginarse, Maxima no resuelve todo. 
Incluso en las ecuaciones más “sencillas”, los polinomios, 
se presenta el primer problema: no hay una fórmula en términos
algebraicos para obtener las raíces de un polinomio de grado 5 
o más. Pero no hay que ir tan lejos. Cuando añadimos raíces, 
logaritmos, exponenciales, etc., la resolución de ecuaciones 
se complica mucho. En esas ocasiones lo más que podemos hacer 
es ayudar a Maxima a resolverlas.

(%i46) eq:x+3=sqrt(x+1);

(%o46) 3+x= 1+x

(%i47) solve(eq,x);

(%o47) [x=-3+ 1+x ]

(%i48) solve(eq^2);

(%o48) [x=-
5+ 7 %i

2
,x=

-5+ 7 %i

2
]

 También podemos resolver sistemas de ecuaciones. Sólo tenemos que 
escribir la lista de ecuaciones y de incógnitas. Por ejemplo:

(%i49) solve([x^2+y^2=1,(x-2)^2+(y-1)^2=4],[x,y]);

(%o49) [[x=
4

5
,y=-

3

5
],[x=0,y=1]]

 Si la solución depende de un parámetro o varios, Maxima utilizará 
%r1, %r2,... para refererirse a estos. Por ejemplo,

(%i50) solve([x+y+z=3,x-y=z],[x,y,z]);

(%o50) [[x=
3

2
,y=-

-3+2 %r1

2
,z=%r1]]

 ¿Qué pasa si el sistema de ecuaciones no tiene solución? Veamos 
un ejemplo:

(%i51) solve([x+y=0,x+y=1],[x,y]);

(%o51) []

  2.2 Resolución aproximada de ecuaciones polinómicas

 Las ecuaciones polinómicas se pueden resolver de manera aproximada. 
Los comandos allroots (que nos da todas las raíces) y realroots 
(que solo nos da las raíces reales) están especializados en encontrar 
soluciones racionales aproximadas de polinomios en una variable:

(%i52) eq:x^9+x^7-x^4+x$
allroots(eq);

(%o53) [x=0.0,x=0.84406777982779+0.30190507748312 %i,x=

0.84406777982779-0.30190507748312 %i,x=-0.32846441923836+

0.89231329168876 %i,x=-0.32846441923836-0.89231329168876 %i,x=-

0.80986929589483+0.5110407920843 %i,x=-0.80986929589483-0.5110407920843

%i,x=0.2942659353054+1.189238256723473 %i,x=0.2942659353054-

1.189238256723473 %i]

 y si solo nos interesan las que son reales:

(%i54) realroots(eq);

(%o54) [x=0]

 3 Análisis Complejo I: Funciones analíticas

 Con wxMaxima podemos calcular directamente algunos límites de 
funciones complejas. En concreto, vamos a poder resolver límites
siempre que la función venga dada en términos de z y la misma
sea analítica (ya que en este caso usamos, básicamente, los mismos
procesos que en el caso de funciones con una variable).
Así, vamos a poder calcular límites de la forma siguiente (es decir,
límites que existen)

 EJEMPLO 1: Calcular los siguientes límites:

-->  remvalue(all);

(%i55) limit(z/(z+1), z, 1);

(%o55) 
1

2

(%i56) limit((%i*z+3)/(z+1), z, -1);

(%o56) infinity

 NOTA: Para el caso que no exista el límite, o también siempre
que la variable z venga expresada en términos de x+%iy, será más
aconsejable expresar la función f(z) en la forma
f(z)=u(x,y)+%iv(x,y)
y calcular (como límites de funciones de dos variables) tanto el
límite de u(x,y) como el de v(x,y).
Lo vemos en el siguiente ejemplo:

 EJEMPLO 2: Calcular el siguiente límite:

(%i57) limit((conjugate(z)+%i*z^2)/cabs(z), z, 0);

(%o57) und

 Observamos que nos da un mensaje de error ya que limit es una función 
de una sola variable, y en este caso, al llevar la función de la que se
quiere calcular su límite tanto un conjugado como un valor absoluto 
(módulo), ésta no va a ser analítica, por lo que para obtener su
límite tendremos que expresarla como dos funciones de dos variables
reales (parte real u(x,y) y parte imaginaria v(x,y)):

(%i58) z:x+%i*y;

(%o58) x+%i y

(%i59) (conjugate(z)+%i*z^2)/cabs(z);

(%o59) 
x-%i y+%i( )x+%i y

2

x
2
+y

2

(%i60) expand(%);

(%o60) 
x

x
2
+y

2
+

%i x
2

x
2
+y

2
-

%i y

x
2
+y

2
-

2 x y

x
2
+y

2
-

%i y
2

x
2
+y

2

(%i61) f:rectform(%);

(%o61) 
x

x
2
+y

2
-

2 x y

x
2
+y

2
+%i









 
  

x
2

x
2
+y

2
-

y

x
2
+y

2
-

y
2

x
2
+y

2

 y elegimos la parte real u(x,y) y la parte imaginaria v(x,y):

(%i62) u:realpart(f);

(%o62) 
x

x
2
+y

2
-

2 x y

x
2
+y

2

(%i63) v:imagpart(f);

(%o63) 
x
2

x
2
+y

2
-

y

x
2
+y

2
-

y
2

x
2
+y

2

 Entonces, y usando las técnicas de cálculo de límites dobles estudiadas
en 1er curso, podremos probar la no existencia de los límites de u(x,y)
y/o v(x,y) (a través de límites direccionales o mediante un cambio a
polares). También puede verse como hacerlo con wxMaxima en la página 
111 del manual que se encuentra en la web:
http://es.scribd.com/doc/52510808/Manual-wxMaxima

 Con wxMaxima también podemos estudiar cuando una función compleja es
analítica, ya que básicamente consiste en aplicar las ecuaciones de
Cauchy-Riemann, que conlleva el cálculo de derivadas parciales de las 
funciones parte real u(x,y) y parte imaginaria v(x,y):

 EJEMPLO 3. Estudiar si la función f(z)=abs(z)^2 admite derivada:

(%i64) remvalue(all);

(%o64) [mi_ecuacion,eq,z,f,u,v]

(%i65) z:x+%i*y$
cabs(z)^2;

(%o66) x
2
+y

2

(%i67) u:realpart(cabs(z)^2);
v:imagpart(cabs(z)^2);

(%o67) x
2
+y

2

(%o68) 0

(%i69) /*calculamos d/dx(u) y d/dy(v)*/
diff(u,x);
diff(v,y);

(%o69) 2 x
(%o70) 0

(%i71) /*calculamos d/dx(v) y -d/dy(u)*/
diff(v,x);
diff(u,y)*(-1);

(%o71) 0

(%o72) -2 y

 Observamos que no se verifican las ecuaciones de C-R salvo si x=y=0.
Por tanto, f'(z) no existe salvo si z=0.

 EJEMPLO 4. Estudiar si la función f(z)=exp(z) admite derivada:

(%i78) remvalue(all);

(%o78) [u,v]

(%i79) z:x+%i*y$

(%i80) exp(z);
rectform(%);

(%o80) %ex+
%i y

(%o81) %ex cos( )y +%i %e
x sin( )y

(%i82) u:realpart(exp(z));
v:imagpart(exp(z));

(%o82) %ex cos( )y

(%o83) %ex sin( )y

(%i84) /*calculamos d/dx (u) y d/dy (v)*/
diff(u,x);
diff(v,y);

(%o84) %ex cos( )y

(%o85) %ex cos( )y

(%i86) /*calculamos d/dx (v) y -d/dy (u)*/
diff(v,x);
diff(u,y)*(-1);

(%o86) %ex sin( )y

(%o87) %ex sin( )y

 Observamos que se verifican las ecuaciones de C-R, por lo que f(z)
es derivable. Además sabemos que f'(z)= diff(u,x)+%i*diff(v,x):

(%i88) derivada_f:diff(u,x)+%i*diff(v,x);

(%o88) %ex cos( )y +%i %e
x sin( )y

 Hemos llegado al conocido resultado de que si f(z)=exp(z), 
entonces f'(z)=f(z). A esta misma conclusión podríamos haber
llegado si hubiésemos realizado

(%i89) remvalue(all)$
/*Primero hemos borrado todas las variables antes de calcular
la derivada tomando z como una variable*/
diff(exp(z),z);

(%o90) %ez

 De nuevo en este ejemplo hemos calculado una derivada como si la
función a derivar no fuese una función compleja, sino que fuese
una función con una sola variable. Esto siempre puede hacerse en
caso que la función a derivar sea analítica, mientras que si no lo
es (porque, por ejemplo, lleva un módulo o un conjugado en su 
definición) no lo vamos a poder hacer (como hemos visto en el 
Ejemplo 3 anterior).

 Otro típico ejercicio de variable compleja relacionado con la derivación
consiste en dada la parte real u(x,y) de una función analítica, hallar 
su función armónica conjugada (es decir, otra función v(x,y) tal que
la función dada por f(z)=u(x,y)+iv(x,y) es analítica). Lo vemos en el 
siguiente ejemplo.

 EJEMPLO 5. Dada la función u(x,y)=3*x^2*y+2*x^2-y^3-2*y^2, probar
que es armónica y hallar su armónica conjugada:

(%i91) remvalue(all)$

(%i92) u(x,y):=3*x^2*y+2*x^2-y^3-2*y^2;

(%o92) u( )x,y :=3 x
2
y+2 x

2
-y

3
+( )-2 y

2

(%i93) diff(u(x,y),x,2)+diff(u(x,y),y,2);

(%o93) 0

 Como este último resultado da 0, es cierto que u(x,y) es armónica.
Para hallar la armónica conjugada, como sabemos que d/dy(v)=d/dx(u),
entonces v(x,y)=int(d/dx(u),respecto y)+g(x)

(%i94) diff(u(x,y),x);

(%o94) 4 x+6 x y

(%i95) integrate(diff(u(x,y),x), y);

(%o95) 4 x y+3 x y
2

 Este último resultado nos da el valor para v(x,y), salvo una cte 
(que es una función que depende de x, g(x)). Usaremos la segunda de las
condiciones de Cauchy-Riemann para calcular g(x). Por la segunda
de estas ecuaciones se tiene que d/dx(v)=-d/dy(u). Calculamos primero
d/dx(v), para lo que derivaremos respecto de x el último resultado 
obtenido anteriormente:

(%i96) diff(integrate(diff(u(x,y),x), y),x,1);

(%o96) 4 y+3 y
2

 Aquí tendremos que sumarle a la expresión anterior g'(x).
El segundo miembro de la igualdad d/dx(v)=-d/dy(u) también lo 
obtenemos haciendo

(%i97) diff(u(x,y),y)*(-1);

(%o97) -3 x
2
+4 y+3 y

2

 y para obtener g'(x) hacemos

(%i98) diff(u(x,y),y)*(-1)-diff(integrate(diff(u(x,y),x), y),x,1);

(%o98) -3 x
2

 e integrando este último resultado obtendremos g(x)

(%i99) integrate(%, x);

(%o99) -x
3

 Al final resulta que v(x,y)=3*x*y^2+4*x*y-x^3.

 4 Análisis Complejo II: Integración

 Como hemos visto en teoría, las integrales en el campo complejo son
siempre integrales de línea en el plano, por lo que podremos usar todo
lo visto en la práctica 2 (tanto a la hora de representar gráficos de
las curvas como a la hora de calcular las integrales de línea). No 
obstante, existen unos resultados específicos para estas integrales
complejas que nos permitirán calcular las mismas sin necesidad de 
acudir a la definición (es decir, sin necesidad de tener que tomar 
una parametrización de la curva plana y de resolver la posterior 
integral de línea). Estos resultados son la independencia del camino 
de integración (más conocida como regla de Barrow para estas 
integrales), y, para el caso de curvas cerradas, el teorema de 
Cauchy-Goursat y las fórmulas integrales de Cauchy (todos estos 
resultados tienen que ver con que f(z) sea analítica en la región 
encerrada por la curva). Intentaremos desarrollar todo ésto en los 
ejemplos que vienen a continuación.

 EJEMPLO 6. Representar las curvas complejas abs(z)=2; abs(z-3)=2:

(%i100) remvalue(all)$

(%i101) load(draw);
(%o101) 

C:/PROGRA~1/MAXIMA~1.0-2/share/maxima/5.28.0-2/share/draw/draw.lisp

(%i103) /*la curva abs(z)=2 equivale a la circunferencia de ecuación
x^2+y^2=4, que es la que representamos a continuación*/

wxdraw2d(implicit(x^2+y^2=4,x,-2,2,y,-2,2));

(%t103) 

(%o103) 

(%i105) /*la curva abs(z-3)=2 equivale a la circunferencia de ecuación
(x-3)^2+y^2=4, que es la que representamos a continuación*/

wxdraw2d(implicit((x-3)^2+y^2=4,x,0,6,y,-2,2));

(%t105) 

(%o105) 

 EJEMPLO 7. Representar las curvas complejas z=2*exp(i*theta);
z=3+2*exp(i*theta) (en este caso las curvas vienen dadas por sus
coordenadas polares; nosotros las representaremos en coordenadas
paramétricas):

(%i106) /*la curva z=2*exp(i*theta) equivale a la parametrización
(2*cos(theta),2*sin(theta)), que es la que representamos 
a continuación*/

wxplot2d([[parametric, 2*cos(theta), 2*sin(theta), 
[theta, 0, 2*%pi], [nticks, 300]]])$

(%t106) 



P1__2017.wxm 13 / 20

(%i107) /*la curva z=3+2*exp(i*theta) equivale a la parametrización
(3+2*cos(theta),2*sin(theta)), que es la que representamos 
a continuación*/

wxplot2d([[parametric, 3+2*cos(theta), 2*sin(theta), 
[theta, 0, 2*%pi], [nticks, 300]]])$

(%t107) 

 Notemos que las dos curvas de este Ejemplo 7 coinciden con las del
Ejemplo 6.

 EJEMPLO 8. Calcular, usando la definición de integral compleja, la
integral de f(z)=conjugate(z) a lo largo de la parte derecha de la
circunferencia abs(z)=2:

(%i108) /*vamos a calcular la integral de f(z)=conjugate(z) a lo largo
de la mitad derecha del círculo abs(z)=2*/

/*La parametrización de la curva viene dada por 
curva(t)=(2cos(t),2sin(t)), donde t varia en [-%pi/2,%pi/2]*/

wxplot2d([[parametric, 2*cos(t), 2*sin(t), 
[t, -%pi/2,%pi/2], [nticks, 300]]])$

(%t108) 

(%i109) /*La función a integrar será f(curva(t))*d/dt(curva(t)) en el
intervalo [-%pi/2,%pi/2]*/

curva(t):=2*cos(t)+%i*2*sin(t);

(%o109) curva( )t :=2 cos( )t +%i 2 sin( )t

(%i110) curva(t)*diff(curva(t),t,1);

(%o110) ( )2 %i cos( )t -2 sin( )t ( )2 cos( )t +2 %i sin( )t

(%i111) integrate(%, t, -%pi/2, %pi/2);
(%o111) 0

 EJEMPLO 9: Vamos a calcular la integral de f(z)=z^2 entre los puntos 
z1=0 y z2=1+i. Lo vamos a hacer:
a) A través de la recta que une ambos puntos.
b) A través de la curva y=x^3 (que también pasa por ambos puntos).
c) A través de la regla de Barrow.

 Para el apartado (a):

(%i112) /*La parametrización de la recta viene dada por 
curva(t)=(t,t)), donde t varia en [0,1]*/
/*La función a integrar será f(curva(t))*d/dt(curva(t)) en el
intervalo [0,1]*/

curva(t):=t+%i*t;

(%o112) curva( )t :=t+%i t

(%i113) (curva(t))^2*diff(curva(t),t);

(%o113) ( )1+%i ( )t+%i t
2

(%i114) integrate(%, t, 0, 1);

(%o114) 
2 %i( )1+%i

3

 Para el apartado (b):

(%i115) /*La parametrización de la curva viene dada por 
curva(t)=(t,t^3)), donde t varia en [0,1]*/

/*La función a integrar será f(curva(t))*d/dt(curva(t)) en el
intervalo [0,1]*/

curva(t):=t+%i*t^3;

(%o115) curva( )t :=t+%i t
3

(%i116) (curva(t))^2*diff(curva(t),t);

(%o116) ( )1+3 %i t
2
( )t+%i t

3 2

(%i117) integrate(%, t, 0, 1);

(%o117) 
-2+2 %i

3

 Observamos que el resultado de (a) y (b) es el mismo a pesar de cambiar
de curva. Esto es debido a que la función f(z)=z^2 es analítica, como
podemos comprobar ya que verifica las ecuaciones de Cauchy-Riemann:

(%i118) z:x+%i*y;
(%o118) x+%i y

(%i119) u:realpart(z^2);
v:imagpart(z^2);

(%o119) x
2
-y

2

(%o120) 2 x y

(%i121) /*calculamos d/dx(u) y d/dy(v)*/
diff(u,x);
diff(v,y);

(%o121) 2 x
(%o122) 2 x

(%i123) /*calculamos d/dx(v) y -d/dy(u)*/
diff(v,x);
diff(u,y)*(-1);

(%o123) 2 y
(%o124) 2 y

 Por tanto también podremos calcular esta integral como nos pide el
apartado (c), es decir, usando la regla de Barrow: Maxima no calcula
integrales definidas si los límites de integración son complejos,
por lo que procederemos a calcular la primitiva de f(z) y evaluarla
en los puntos 0 y 1+i:

(%i125) remvalue(all)$
integrate(z^2, z);

(%o126) 
z
3

3

(%i127) f(z):=z^3/3;

(%o127) f( )z :=
z
3

3

(%i128) f(1+%i)-f(0);
rectform(%);

(%o128) 
( )1+%i

3

3

(%o129) -
2

3
+
2 %i

3

 5 Análisis Complejo III: Series. Ceros y polos.

Residuos

 wxMaxima nos permite obtener los ceros y los polos (que sabemos
son conceptos relacionados) de funciones complejas. También podremos
obtener el valor de los residuos en los polos.

 Los ceros de una función compleja no son sino las raíces que tiene
la propia función; mientras que los polos suelen ser las raíces del
denominador (siempre que no anulen el numerador, porque entonces
puede ser que tengamos una singularidad evitable). Veamos un ejemplo:

 EJEMPLO 10: Vamos a definir la función 
f(z)=(z^2+z-1)/(z^3-1.5*z^2-0.5*z-1)
y ver como podemos elegir su numerador y denominador:

(%i130) remvalue(all)$
f(z):=(z^2+z-1)/(z^3-1.5*z^2-0.5*z-1);

(%o131) f( )z :=
z
2
+z-1

z
3
-1.5 z

2
+( )-0.5 z-1

(%i132) num(f(z));

(%o132) -1+z+z
2

(%i133) denom(f(z));

(%o133) -1-0.5 z-1.5 z
2
+z

3

 EJEMPLO 11: Si queremos calcular los ceros de 
f(z)=(z^2+z-1)/(z^3-1.5*z^2-0.5*z-1)
haremos

(%i134) remvalue(all)$
f(z):=(z^2+z-1)/(z^3-1.5*z^2-0.5*z-1);

(%o135) f( )z :=
z
2
+z-1

z
3
-1.5 z

2
+( )-0.5 z-1

(%i136) solve([f(z)=0], [z]);

rat: replaced -0.5 by -1/2 = -0.5

rat: replaced -1.5 by -3/2 = -1.5

(%o136) [z=-
1+ 5

2
,z=

-1+ 5

2
]

 mientras que para calcular las singularidades (polos) haremos

(%i137) solve([z^3-1.5*z^2-0.5*z-1=0], [z]);

rat: replaced -0.5 by -1/2 = -0.5

rat: replaced -1.5 by -3/2 = -1.5

(%o137) [z=-
1+ 7 %i

4
,z=

-1+ 7 %i

4
,z=2]

 Como observamos que no hay repetición entre las raíces del numerador
y denominador, tanto los ceros como los polos son simples.

 Si queremos comprobar si estos polos están, por ejemplo, en el 
interior del círculo unidad, obtendremos su módulo y preguntamos
si éste es menor que la unidad:

(%i138) cabs(-(sqrt(7)*%i+1)/4);

(%o138) 
1

2

(%i139) is(%<1);
(%o139) true

 Y lo mismo para el segundo polo:

(%i140) cabs((sqrt(7)*%i-1)/4);

(%o140) 
1

2

(%i141) is(%<1);
(%o141) true

 mientras que para el tercero:

(%i142) cabs(2);
is(%<1);

(%o142) 2
(%o143) false

 Por tanto, los dos primeros sí que están en el interior del círculo
unidad, mientras que el tercero (z=2) está fuera.

 Podemos desarrollar una función en Serie de potencias en torno a un 
punto z0 (es decir, desarrollar f como potencias de z-z0):

 EJEMPLO 12: Si queremos desarrollar como potencias de z (es decir, 
tomando z0=0) la función 
g(z)=(z+1)/(z^3+4*z^2-3*z-18)
haremos

(%i144) remvalue(all)$
g(z):=(z+1)/(z^3+4*z^2-3*z-18);

(%o145) g( )z :=
z+1

z
3
+4 z

2
+( )-3 z-18

(%i146) powerseries(g(z),z,0);

(%o146) 

i1=0

∞

-
3 2

-1-i1
z
i1

25
+
2( )1+i1 3

-2-i1
( )-1

i1
z
i1

5
-
( )-1

i1
z
i1

25 3
i1

(%i147) niceindices(%);

(%o147) 

i=0

∞

-
3 2

-1-i
z
i

25
+
2( )1+i 3

-2-i
( )-1

i
z
i

5
-
( )-1

i
z
i

25 3
i

 Podemos desarrollar una función en Serie de Laurent en torno a un 
punto z0 (es decir, desarrollar f como potencias de z-z0, pudiendo
ser los exponentes de z-z0 positivos o negativos):

 EJEMPLO 13: Si queremos desarrollar como serie de Laurent en potencias de 
z+3 (ya que -3 es un polo de dicha función) la anterior función g(z) 
haremos

(%i148) niceindicespref;
niceindicespref:[n,m]$
niceindices(powerseries(f(z),z,-3));

(%o148) [i,j,k,l,m,n]

rat: replaced 5.0 by 5/1 = 5.0

rat: replaced -40.0 by -40/1 = -40.0

rat: replaced 35.5 by 71/2 = 35.5

rat: replaced -10.5 by -21/2 = -10.5

rat: replaced 1.0 by 1/1 = 1.0

rat: replaced -5.0 by -5/1 = -5.0

rat: replaced -40.0 by -40/1 = -40.0

rat: replaced 35.5 by 71/2 = 35.5

rat: replaced -10.5 by -21/2 = -10.5

rat: replaced 1.0 by 1/1 = 1.0

rat: replaced 1.0 by 1/1 = 1.0

rat: replaced -40.0 by -40/1 = -40.0

rat: replaced 35.5 by 71/2 = 35.5

rat: replaced -10.5 by -21/2 = -10.5

rat: replaced 1.0 by 1/1 = 1.0
(%o150) 

n=0

∞ 









-
2

11 5
n
-

( )-11+ 7 %i
-1-n 






  10+

5( )-11+ 7 %i

2
+
( )-11+ 7 %i

2

8
( )-4

1+n

71+
21( )-11+ 7 %i

2
+
3( )-11+ 7 %i

2

8

-

( )11+ 7 %i
-1-n 


10-

5(11

71-
21(11+ 7 %i

2

 Una vez que conocemos los polos de una función, podemos calcular de manera
sencilla sus residuos en dichos polos:

 EJEMPLO 14: Si queremos calcular los residuos de la anterior función g(z) 
en sus puntos singulares (que son 2 y -3):

(%i151) residue(g(z),z,2);

(%o151) 
3

25

(%i152) residue(g(z),z,-3);

(%o152) -
3

25

 

 6 Instrucciones para la entrega de las prácticas

 - Las prácticas hay que hacerlas en fichero Maxima (extensión .wxm) 
y remitirlas por email.
- Aconsejable realizar un fichero para cada práctica (llamandolo, 
por ejemplo, Practica 1-nombre alumno.wxm)
- En los ficheros ir explicando lo que se va a realizar, incluyendo
los comentarios que sean necesarios, como en el ejemplo de la sección
anterior, y que siempre van en la forma:
/*COMENTARIO QUE SE QUIERE HACER*/
- Comprobar que no hay erratas y las sentencias escritas son correctas.



P1__2017.wxm 19 / 20

 7 Un ejemplo resuelto

 Lo que viene a continuación es una muestra de lo que hay que remitir
en el fichero de prácticas. Copiar en primer lugar el enunciado y poner
la resolución del ejercicio a continuación (observar como se van 
introduciendo los comentarios que aclaran lo que estamos realizando
en la resolución).

 PROBLEMA Nº. Resolver, con ayuda de wxMaxima, el apartado (b)
del PROBLEMA 4 del examen de Matemáticas II de la convocatoria 
de JUNIO 2014, justificando lo que se realice.

(%i153) /*Se trata de obtener la integral compleja de la función 
f(z)=((z+1)*exp(z))/(z^7*(z^7-%pi*%i)) en el círculo unidad*/
/*Esta función tiene por punto singular aislado en el círculo
unidad solamente el punto z0=0, que será un polo de orden 7.
Por tanto, usando el teorema de los residuos sabemos que el 
valor de la integral coincide con 2*%pi*%i*residue(f(z),z,0)*/
/*¿Como obtener residue(f(z),z,0)? Si tuviésemos que calcularlo
a mano, deberíamos de actuar como en la resolución del examen, 
realizando un desarrollo de Laurent como potencias de z para
f(z). Sin embargo, wxMaxima nos evita todos estos cálculos,
puesto que nos calcula el residuo directamente, como vemos a
continuación*/
/*Introducimos f(z)*/

remvalue(all)$
f(z):=((z+1)*exp(z))/(z^7*(z^7-%pi*%i));

(%o154) f( )z :=
( )z+1 exp( )z

z
7
( )z

7
-π %i

(%i155) /*Y directamente aplicamos el teorema de los residuos (no es
preciso resolver la ecuación del denominador, ya que trivial-
mente se observa que el único punto singular aislado en el 
interior de la circunferencia unidad es z0=0)*/

/*Por tanto el valor de la integral viene dado por*/

2*%pi*%i*residue(f(z),z,0);

(%o155) -
7

360

 8 Ejercicios a entregar

 PROBLEMA 1. Realizar las siguientes operaciones elementales, siendo A la
última cifra no nula del NIF del alumno, B la penúltima, ...:
a) Si z1=A+B*%i, z2=C-D*%i y z3=-A/B+%i*(3^(1/2))/2, calcular
abs(C*z1-D*z2), (z3)^(-A) y (z2)^(1/4).
b) Expresar en forma polar el complejo z=A+%i*B*(C^(1/2)).
c) Calcular todas las raíces del polinomio p(z)=z^3+A*z^2+B*z+C=0

 PROBLEMA 2. Resolver, con ayuda de wxMaxima, el PROBLEMA 5 
del examen de Matemáticas II de la convocatoria de JUNIO 2016,
justificando lo que se realice.

 PROBLEMA 3. Resolver, con ayuda de wxMaxima, el PROBLEMA 5, apdo b, 
del examen de Matemáticas II de la convocatoria de FEBRERO 2016, 
justificando lo que se realice.


